TOV equations for structure of spherical star

In Misner-Thorne-Wheeler texbook we have the following equations

$$\frac{dp}{dr} = -\frac{(\rho + p)(m + 4\pi r^3 p)}{r(r-2m)}$$
$$\frac{dm}{dr} = 4\pi r^2 \rho$$
$$\frac{d\phi}{dr} = \frac{m + 4\pi r^3 p}{r(r-2m)}$$

Here $\rho = \rho_0 + \epsilon$ where $\rho_0 = n_\mathrm{b} m_\mathrm{b}$ is rest-mass density and ϵ internal energy. We assume polytropic equation of state (see eqs. (1.74) a (1.86) in [Baumgarte&Shapiro])

$$p = k\rho_0^\gamma,$$

i.e.

$$\rho_0 = (p/k)^{1/\gamma} \quad \text{and} \quad \rho = \rho_0 + \frac{p}{\gamma - 1}$$

We put $k = 1$ and consider only dimensionless problem (see [Baumgarte&Shapiro] for scaling of true quantities such as stellar radius or m for $k \neq 1$).

We get only inner part of the metric

$$ds^2 = -e^{2\Phi(r)}dt^2 + \frac{dr^2}{1 - \frac{2m(r)}{r}} + r^2 d\Omega^2$$

Note that the code does not solve function $\Phi(r)$ completely, because in the first step we simply assume $\Phi(r=0) = 0$ which is not OK, tl right boundary condition is $\Phi(\infty) = 0$ or $e^{2\phi(r_s)} = 1 - 2m(r_s)/r_s$ at surface of the star.

In[1]:=
```
γ = 2.7
```
Out[1]=
```
2.7
```

Here the right-hand side of differential equations is defined

In[2]:=
```
ρ[p_Real] := Abs[p]^(1 / γ)
μ[p_Real] := HeavisideTheta [p] ( ρ[p] + p / (γ - 1))
dpdr[r_Real , p_Real , m_Real] := If[r > 0, -( μ[p] + p) (m + 4 Pi r ^ 3 p) / (r (r - 2 m)), 0]
```

In[5]:=
```
eqs = {
   p '[r] == dpdr[r , p[r], m[r]],
   m '[r] == 4 Pi r ^ 2 μ[p[r]]
  }
```

Out[5]=
```
{p′[r] == dpdr[r , p[r], m[r]], m′[r] == 4 π r^2 μ[p[r]]}
```

We solve it as initial-value ODE problem and ignore ϕ(r)

In[6]:=
```
sol = NDSolve[ eqs && {p[0] == 1, m[0] == 0} && WhenEvent [p[r] < 0 , "StopIntegration "] &&
    WhenEvent [ r - 2 m[r] < 0, "StopIntegration "] , {p[r], m[r]}, {r , 0, 10}]
```

Out[6]=
```
{{p[r] → InterpolatingFunction [ Domain : {{0., 0.406}} Output : scalar ][r],
  m[r] → InterpolatingFunction [ Domain : {{0., 0.406}} Output : scalar ][r]}}
```

The results are plotted

In[7]:=
```
Plot[1 - 2 m[r] / r /. sol, {r, 0, sol[[1, 1, 2, 0]]["Domain"][[1, 2]]}, PlotTheme → "Detailed"]
Plot[p[r] /. sol, {r, 0, sol[[1, 1, 2, 0]]["Domain"][[1, 2]]}, PlotTheme → "Detailed"]
```

Out[7]=



Out[8]=



To plot mass-radius relation we need following function

In[29]:=
```
rMaxM[p0_Real] := Block[{sol, rMax},
   sol =
    NDSolve[eqs && {p[0] == p0, m[0] == 0} && WhenEvent[p[r] < 0 , "StopIntegration"] &&
      WhenEvent[r - 2 m[r] < 0, "StopIntegration"], {p[r], m[r]}, {r, 0, 10}];
   rMax = sol[[1, 1, 2, 0]]["Domain"][[1, 2]];
   {rMax, sol[[1, 2, 2, 0]][rMax]}
  ]
rMaxM[1.]
```

Out[30]= {0.405512, 0.126578}

In[10]:=
```
ParametricPlot[rMaxM[Exp[logpC]], {logpC, -15, 15},
 PlotPoints → 30, MaxRecursion → 1, PlotRange → Full,
 AspectRatio → 2/3, PlotTheme → "Detailed", FrameLabel → {r, m}]
```

Out[10]=